This maple file illustrates the use of the procedure *DarbouxPolynomials* of the package *RationalFirstIntegrals* writen by *A. Bostan, G. Chèze, T. Cluzeau, and J.-A. Weil.*
For explanations and details about the algorithms behind the procedure, see our related paper:

Efficient algorithms for computing rational first integrals of planar polynomial vector fields.

```
> restart:
> with(RationalFirstIntegrals);
```

[*DarbouxPolynomials*, *DeterministicRationalFirstIntegral*, *GenericRationalFirstIntegral*, *GuessMinimalPolynomial*, *HeuristicRationalFirstIntegral*, *PadeHermiteHeuristic*, *ProbabilisticRationalFirstIntegral*, *SimplifyRFI*, *simpli1*, *simpli2*] **(1)**

Consider the jacobian derivation associated with the polynomial:

```
> f:=(y*x-1)*(x-y^2)*(y-x-1);
```

$$f := (x\,y-1)\left(-y^2+x\right)(y-x-1) \quad (2)$$

Namely, the planar polynomial vector field {x'=A(x,y), y'=B(x,y)} by giving the polynomials A and B:

```
> A:=simplify(-diff(f,y));
  B:=simplify(diff(f,x));
```

$$A := -3\,x^2\,y^2+4\,x\,y^3+x^3-2\,x^2\,y-3\,x\,y^2+x^2+2\,x\,y-3\,y^2+x+2\,y$$
$$B := 2\,x\,y^3-y^4-3\,x^2\,y+2\,x\,y^2+y^3-2\,x\,y-y^2+2\,x-y+1 \quad (3)$$

We define the associated derivation:

```
> DD:=proc(f) A*diff(f,x)+B*diff(f,y) end:
```

Let us search for all irreducible Darboux polynomials of the derivation DD of degree at most 3.

We need a point *pt* for which A(pt,y) <> 0:

```
> pt:=0;
  check_pt:=eval(A,x=pt);
```

$$pt := 0$$
$$check\_pt := -3\,y^2+2\,y \quad (4)$$

We first check using for example the ProbabilisticRationalFirstIntegral procedure that it does not admit a rational first integral of degree at most 3.

```
> ProbabilisticRationalFirstIntegral(A,B,pt,[-1,1],3);
```

"None" **(5)**

We run our Darboux polynomial procedure:

```
> DP:={DarbouxPolynomials(A,B,pt,3)};
```

$$DP := \{y-x-1\} \quad (6)$$

We check that it is a Darboux polynomial.

```
> Lambda:=normal(DD(DP[1])/DP[1]);
```

$$\Lambda := -3\,x\,y^2-y^3+x^2+2\,x\,y+2\,y-1 \quad (7)$$

We find only one irreducible Darboux polynomial. However by construction, we know that the two other factors of f are also Darboux polynomials so that we miss them.

The reason why we miss these two Darboux polynomials is that we are not in a "generic" situation (for more explanations, see our paper).
Let us make a change of coordinates in order to be in a "generic situation".

```
> a:=2;
```

$$a := 2 \qquad (8)$$

```
> Aa:=simplify(subs(x=x+a*y,A)-a*subs(x=x+a*y,B));
```

$$Aa := -3\,x^2\,y^2 - 12\,x\,y^3 - 10\,y^4 + x^3 + 10\,x^2\,y + 21\,x\,y^2 + 8\,y^3 + x^2 + 10\,x\,y + 15\,y^2 - 3\,x - 2\,y - 2 \qquad (9)$$

```
> Ba:=simplify(subs(x=x+a*y,B));
```

$$Ba := 2\,x\,y^3 + 3\,y^4 - 3\,x^2\,y - 10\,x\,y^2 - 7\,y^3 - 2\,x\,y - 5\,y^2 + 2\,x + 3\,y + 1 \qquad (10)$$

```
> DPa:={DarbouxPolynomials(Aa,Ba,pt,3)};
```

$$DPa := \left\{1 + x + y, -\frac{1}{2} + \frac{1}{2}\,x\,y + y^2, y^2 - x - 2\,y\right\} \qquad (11)$$

Now we correctly find 3 irreducible Darboux polynomials of degree at most 3.

We perform the inverse change of coordinates to get all irreducible Darboux polynomials of the original polynomial vector field.

```
> DPgood:={seq(simplify(subs(x=x-a*y,DPa[i])),i=1..nops(DPa))};
```

$$DPgood := \left\{-\frac{1}{2} + \frac{1}{2}\,x\,y, 1 + x - y, y^2 - x\right\} \qquad (12)$$

We can chech that they are Darboux polynomials by computing their cofactors.

```
> Lambdagood:={seq(normal(DD(DPgood[i])/DPgood[i]),i=1..3)};
```

$$Lambdagood := \{-x\,y^2 + 3\,y^3 - 2\,x^2 - 2\,y^2 - x, -3\,x\,y^2 - y^3 + x^2 + 2\,x\,y + 2\,y - 1, 4\,x\,y^2 - 2\,y^3 + x^2 - 2\,x\,y + 2\,y^2 + x - 2\,y + 1\} \qquad (13)$$